
Factor Analysis in SPSS Using a Correlation Matrix as Input 

In SPSS, factor (or component) analysis can be estimated by inputting a correlation (or 

covariance) matrix rather than the raw data as input. This can be useful when rerunning 

an analysis from data supplied in an article or a report. 

To read in a correlation matrix for input to SPSS, use syntax such as that shown below. 

MATRIX DATA VARIABLES=x1 x2 x3 x4 x5 x6/CONTENTS=CORR N. 
BEGIN DATA. 
1.00  
0.70 1.00 
0.65 0.75 1.00 
0.20 0.05 0.25 1.00 
0.15 0.10 0.15 0.65 1.00 
0.10 0.20 0.20 0.75 0.60 1.00 
100 100 100 100 100 100 
END DATA. 

In the syntax above, the command MATRIX DATA indicates that data in the form of a 
matrix (rather than raw data) is to be read in.  

The subcommand VARIABLES = specified the names of the variables. There must be 
as many names as there are variables in the matrix (here, there are 6). 

The subcommand CONTENTS = CORR N specifies that a correlation matrix and a 
vector of sample sizes will be read in. When reading in matrix data, sample sizes must 
be supplied as these cannot be obtained from the correlation matrix, 

The BEGIN DATA command indicates that the data will begin on the next line. 

After the BEGIN DATA command, enter the lower diagonal of the correlation matrix, 
including the 1’s on the diagonal. 

On the line after the correlation matrix, enter the sample sizes. If using listwise deletion, 
these should be the sample size after listwise deletion, and the resulting sample size 
will be the same for each variable. Even though the sample size value is the same for 
each variable, it must be entered for each. Here, I have entered the value “100” for each 
variable. 

The END DATA command indicates the end of matrix data entry. 

Running this set of commands will cause the data window below to appear in SPSS: 

These data can now be used in SPSS analyses such as FACTOR that accept matrix 
data. The usual commands are used with one exception. 
 
In the FACTOR command, include the syntax MATRIX = IN(CORR = *). This indicates 
that matrix data should be used. The specification IN(CORR = *) indicates that the input 
matrix is a correlation matrix and that the current active file (indicated by an asterisk) 
should be used. 
 
FACTOR matrix=in(corr=*) 
  /ANALYSIS x1 x2 x3 x4 x5 x6 
  /PRINT INITIAL CORRELATION REPR EXTRACTION  ROTATION 
  /CRITERIA FACTORS(1) ITERATE(25) 
  /EXTRACTION PAF 
  /ROTATION OBLIMIN 
  /METHOD=CORRELATION . 

